Задание:

1. Дано четное число N (> 0) и символы C1 и C2. Вывести строку длины N, которая состоит из чередующихся символов C1 и C2,

2. Дана строка, изображающая целое положительное число. Вывести сумму цифр этого числа.

3. Дан символ C и строки S, S0. Перед каждым вхождением символа C в строку S вставить строку S0.

Перечисляемый тип данных

Перечисляемый тип представляет собой ограниченную упорядоченную последовательность скалярных констант, составляющих данный тип. Значение каждой константы задается ее именем. Имена отдельных констант отделяются друг от друга запятыми, а вся совокупность констант, составляющих данный перечисляемый тип, заключается в круглые скобки. 

Перечисляемый тип описывается в разделе описания типов, например: 

type
     Rainbow = (RED, ORANGE, YELLOW, 

             GREEN, LIGHT_BLUE, BLUE, VIOLET);

Каждое значение является константой своего типа и может принадлежать только одному из перечисляемых типов, заданных в программе. Например, перечисляемый тип Traffic_Light не может быть определен в одной программе с типом Rainbow, так как оба типа содержат одинаковые константы. 

Описание переменных, принадлежащих к скалярным типам, которые объявлены в разделе описания типов, производится с помощью имен типов. Например: 

type  Traffic_Light= (RED, YELLOW, GREEN);

var   Section: Traffic_Light;

Это означает, что переменная Section может принимать значения RED, YELLOW или GREEN. 

Переменные перечисляемого типа могут быть описаны в разделе описания переменных, например: 

var Section: (RED, YELLOW, GREEN);

При этом имена типов отсутствуют, а переменные определяются совокупностью значений, составляющих данный перечисляемый тип. 

К переменным перечисляемого типа может быть применим оператор присваивания: 

Section:= YELLOW;

Упорядоченная последовательность значений, составляющих перечисляемый тип, автоматически нумеруется, начиная с нуля и далее через единицу. Отсюда следует, что к перечисляемым переменным и константам могут быть применены операции отношения и стандартные функции Pred, Succ, Ord. 

Интервальный тип данных

Отрезок (диапазон значений) любого порядкового типа может быть определен как интервальный (ограниченный) тип. Отрезок задается диапазоном от минимального до максимального значения констант, разделенных двумя точками. В качестве констант могут быть использованы константы, принадлежащие к целому, символьному, логическому или перечисляемому типам. Скалярный тип, на котором строится отрезок, называется базовым типом. Примеры отрезков: 

1..10

-15..25

'a'..'z'

Минимальное и максимальное значения констант называются нижней и верхней границами отрезка, определяющего интервальный тип. Нижняя граница должна быть меньше верхней. 

Над переменными, относящимися к интервальному типу, могут выполняться все операции и применяться все стандартные функции, которые допустимы для соответствующего базового типа. 

Множества в языке Pascal

Множество — это структурированный тип данных, представляющий собой набор взаимосвязанных по какому-либо признаку или группе признаков объектов, которые можно рассматривать как единое целое. Каждый объект в множестве называется элементом множества. 

Все элементы множества должны принадлежать одному из порядковых типов, содержащему не более 256 значений. Этот тип называется базовым типом множества. Базовый тип задается диапазоном или перечислением. 

Область значений типа множество — набор всевозможных подмножеств, составленных из элементов базового типа. В выражениях на языке Паскаль значения элементов множества указываются в квадратных скобках: [1,2,3,4], ['а',‘b','с'], ['a'..'z']. 

Если множество не имеет элементов, оно называется пустым и обозначается как []. Количество элементов множества называется его мощностью. 

Множество может принимать все значения базового типа. Базовый тип не должен превышать 256 возможных значений. Поэтому базовым типом множества могут быть byte, char, boolean и производные от них типы. 

Множество в памяти хранится как массив битов, в котором каждый бит указывает является ли элемент принадлежащим объявленному множеству или нет. Максимальное число элементов множества 256, а данные типа множество могут занимать не более 32 байт. 

Число байтов, выделяемых для данных типа множество, вычисляется по формуле: 

ByteSize = (max div 8) - (min div 8) + 1,
где max и min — верхняя и нижняя границы базового типа данного множества. 

Номер байта для конкретного элемента Е вычисляется по формуле: 

ByteNumber = (E div 8) - (min div 8),
номер бита внутри этого байта по формуле: 

BitNumber = E mod 8
Не имеет значения порядок записи элементов множества внутри конструктора. Например, [1, 2, 3] и [3, 2, 1] — это эквивалентные множества. 

Каждый элемент в множестве учитывается только один раз. Поэтому множество [1, 2, 3, 4, 2, 3, 4, 5] эквивалентно [1..5]. 

Переменные множественного типа описываются так:
Var <идентификатор> : set of <базовый тип>; 

Например: 

           Var A, D : Set Of Byte; 

               B : Set Of 'a'..'z'; 

               C : Set Of Boolean;

Нельзя вводить значения во множественную переменную процедурой ввода и выводить процедурой вывода. 

Множественная переменная может получить конкретное значение только в результате выполнения оператора присваивания:
<множественная переменная> := <множественное выражение>; 

Например: 

A : = [50, 100, 150, 200];

B : = ['m', 'n', 'k'];
C : =[True, False]; 

D : = A;

Кроме того, выражения могут включать в себя операции над множествами. 

Операции над множествами

Объединением двух множеств A и B называется множество, состоящее из элементов, входящих хотя бы в одно из множеств A или B. Знак операции объединения в Паскале «+». 


Примеры: 

1) [1, 2, 3, 4] + [3, 4, 5, 6] => [1, 2, 3, 4, 5, 6]

2) []+[‘a’..’z’]+[‘A’..’E’, ‘k’]  => [‘A’..’E’, ‘a’..’z’]

3) [5<4, true and false] + [true]  => [false, true]

Пересечением двух множеств A и B называется множество, состоящее из элементов, одновременно входящих во множество A и во множество B. Знак операции пересечения в Паскале «*» 

Примеры: 

1) [1, 2, 3, 4] * [3, 4, 5, 6] => [3, 4] 

2) [‘a’..’z’]*[‘A’..’E’, ‘k’]  => [‘k’]

3) [5<4, true and false] * [true]  => []

Разностью двух множеств A и B называется множество, состоящее из элементов множества A, не входящих во множество B. 

Примеры: 

1a) [1, 2, 3, 4] - [3, 4, 5, 6] => [1, 2]

1b)  [3, 4, 5, 6] - [1, 2, 3, 4]  => [5, 6]

2a) [‘a’..’z’]-[‘A’..’E’, ‘k’]  => [‘a’..’j’, ‘i’..’z’]

2b) [‘A’..’E’, ‘k’] - [‘a’..’z’]  => [‘A’..’E’]

3a) [5<4, true and false] - [true] => [false]

3b) [true] - [5<4, true and false] => [true]

Операция вхождения. Это операция, устанавливающая связь между множеством и скалярной величиной, тип которой совпадает с базовым типом множества. Если x — такая скалярная величина, а M — множество, то операция вхождения записывается так: x in M. 

Результат — логическая величина true, если значение x входит в множество M, и false — в противном случае. 

Например, 4 in [3, 4, 7, 9] –– true, 5 in [3, 4, 7, 9] –– false. 

Используя данную операцию, можно не только работать с элементами множества, но и, даже если в решении задачи явно не используются множества, некоторые логические выражения можно записать более лаконично. 

1) Натуральное число n является двухзначным. Вместо выражения (n >= 10) and (n <=99) можно записать n in [10..99]. 

2) Символ c является русской буквой. Вместо выражения (c >= ‘А’) and (c <= ‘Я’) or (c>=‘а’) and (c<=‘п’) or (c>=‘р’) and (c<=‘я’) пишем c in [‘А’.. ‘Я’, ‘а’.. ‘п’, ‘р’.. ‘я’] и т.д. 

Добавить новый элемент в множество можно с использованием операции объединения. Например, a:= a+[5] Для этих же целей в Turbo Pascal  предназначена процедура Include: include (M, A) M – множество, A – переменная того же типа, что и элементы множества M. Тот же пример можно записать так: Include (a, 5) 

Исключить элемент из множества можно с помощью операции «разность множеств». Например, a:= a-[5] Для этих же целей в Turbo Pascal 7.0 предназначена процедура Exclude: exclude (M, A) M – множество, A – переменная того же типа, что и элементы множества M. Тот же пример можно записать так: Exclude (a, 5)

Задача 1. Дана строка. Сохранить в ней только первые вхождения символов, удалив все остальные. 

program ex_set_3;

var m : set of char;

    s : string; i : byte;

begin

    write('Введите строку: ');

    readln(s);

    m :=[];

    i := 1;

    while i <= length(s) do

    if s[i] in m then delete(s, i, 1)

                    else begin m:=m+[s[i]]; i := i + 1 end;

    writeln(s)

end.

 Задача 2.
Дана строка символов. Удалить из нее все знаки препинания.
var   str, str1 : string;

        L, i : integer;

         m : set of char;

begin

         m:=['.', ',', '!', ':', ';', '?', '-'];

         writeln('Введите текст');

         readln(str);  L:=length(str); str1:='';

         for i:=1 to L do

                if not(str[i] in m) then str1:=str1+str[i];

         str:=str1;

         writeln('Преобразованный текст ', str);

end.

 Задача 3. (Для самостоятельного решения)
Дана строка-предложение на русском языке. Подсчитать количество содержащихся в строке гласных букв
Контрольные вопросы и задания
1. Что такое множество?

2. Почему множество является структурированным типом данных?

3. Как хранится множество в памяти ЭВМ? Какой максимальный объем оперативной памяти может быть отведен под хранение одного множества? 

4. Какие операции можно выполнять над множествами? 

5. Как добавить элемент в множество? 

6. Как исключить элемент из множества? 

7. Как вывести элементы множества? Как подсчистать количество элементов в множестве? 

8. Как может быть использована операция вхождения? 

